

16

Getting around with StreetMap USA


Jonathan Bailey




The ESRI® StreetMap™ USA extension provides detailed street data for the entire United States that you can use in your applications. With the StreetMap USA extension, you can use StreetMap data sources just like any other feature class. The StreetMap USA extension extends the geodatabase and ArcCatalog™ object models so that your applications can seamlessly use StreetMap data sources as geodatabase or ArcCatalog objects.

In order to use StreetMap data sources, your application must have a valid StreetMap USA license, and the StreetMap USA extension must be enabled. In order to develop applications that use the StreetMap USA extension, your application must reference the StreetMap USA extension object library.

StreetMap objects



STREETMAPEXTENSION CLASS



The *StreetMapExtension* class enables ArcGIS to read StreetMap data sources as feature data object data sources.

The *StreetMapExtension* class is an *Extension* class that enables ArcGIS™ to read StreetMap data sources as feature data object data sources. Using the *StreetMapExtension* class, you can develop applications that use StreetMap data sources like any other *FeatureClass* object.


In order to be able to read StreetMap data sources, your application must have a valid StreetMap USA license, and the StreetMap USA extension must be enabled.

The following VBA (Visual Basic® for Applications) code can be used to obtain a reference to the *Application* object's *StreetMapExtension* object and to ensure that it is enabled. This code assumes that you have created a reference to the ESRI StreetMap USA extension object library in VBA.

```
Sub EnableStreetMapExtension()
    Dim pESRILicenseInfo As IESRILicenseInfo
    Dim pUID As UID
    Dim pApplication As IApplication
    Dim pExtensionConfig As IExtensionConfig

    Set pESRILicenseInfo = New ESRILicenseInfo
    If Not pESRILicenseInfo.IsLicensed(esriProductCodeStreetMap) Then
        Exit Sub
    End If
    Set pUID = New UID
    pUID.Value = "esricore.StreetMapExtension"
    Set pApplication = ThisDocument.Parent
    Set pExtensionConfig = pApplication.FindExtensionByCLSID(pUID)
    If Not pExtensionConfig.State = esriEEnabled Then
        pExtensionConfig.State = esriEEnabled
    End If
End Sub
```

STREETMAPWORKSPACEFACTORY COCLASS



The *StreetMapWorkspaceFactory* coclass is a *WorkspaceFactory* that you can use to open *StreetMap* Workspaces.


The *StreetMapWorkspaceFactory* coclass is a *WorkspaceFactory* class that you can use to open *StreetMap* *Workspace* objects. A *StreetMap* *Workspace* object contains *StreetMap* *FeatureClass* objects, which are stored as files with a “.edg” extension.

In order to use the *StreetMapWorkspaceFactory* coclass, you must have a valid *StreetMap* USA license, and the *StreetMap* USA extension must be present and enabled. To ensure that your code uses the same *StreetMapExtension* object as the *ArcMap*[™] or *ArcCatalog* *Application* object, use the *IObjectFactory::Create* method on the *Application* object to create the *StreetMapWorkspaceFactory* object.

The following VBA code demonstrates how to open a *StreetMap* *FeatureClass* object using the *StreetMapWorkspaceFactory* coclass. This code assumes that you have a valid *StreetMap* USA license and that the *StreetMap* USA extension is enabled.

```
Sub OpenStreetMapFeatureClass()
    Dim pObjectFactory As IObjectFactory
    Dim pWorkspaceFactory As IWorkspaceFactory
    Dim pFeatureWorkspace As IFeatureWorkspace
    Dim pFeatureClass As IFeatureClass

    Set pObjectFactory = ThisDocument.Parent
    Set pWorkspaceFactory = _
        pObjectFactory.Create("esricore.StreetMapWorkspaceFactory")
    Set pFeatureWorkspace = pWorkspaceFactory.OpenFromFile _
        ("d:\data\streets", 0)
    Set pFeatureClass = pFeatureWorkspace.OpenFeatureClass("usa.edg")
End Sub
```



The *GxStreetMapFactory* coclass is a *GxObjectFactory* class that creates *GxObject* objects that represent StreetMap *FeatureClass* objects. In general, your applications will not need to use the *GxStreetMapFactory* coclass.


The *GxStreetMapDataset* coclass is a *GxObject* class that represents a StreetMap *FeatureClass* object. You can obtain a *GxStreetMapDataset* object from ArcCatalog, then open the StreetMap *FeatureClass* object that it represents.

The following VBA code demonstrates how to retrieve a StreetMap *FeatureClass* object from a *GxStreetMapDataset* object that you have selected in ArcCatalog. This code assumes that you have a valid StreetMap USA license, that the StreetMap USA extension is present and enabled, and that you have selected a StreetMap feature class in ArcCatalog.

```

Sub OpenSelectedStreetMapFeatureClass()
    Dim pGxApplication As IGxApplication
    Dim pGxStreetMapDataset As IGxDataset
    Dim pFeatureClass As IFeatureClass


    Set pGxApplication = ThisDocument.Parent
    Set pGxStreetMapDataset = pGxApplication.SelectedObject
    Set pFeatureClass = pGxStreetMapDataset.Dataset
End Sub
    
```



The *GxStreetMapFactory* coclass extends ArcCatalog by creating *GxObjects* for StreetMap data sources.

The *GxStreetMapDataset* coclass is a *GxObject* that represents a StreetMap data source.

GxFILTERSTREETMAPFEATURECLASSES COCLASS



The *GxFilterStreetMapFeatureClass* coclass can be used to allow users to browse for *StreetMap* FeatureClasses using a *GxDialog*.

The *GxFilterStreetMapFeatureClasses* coclass is a *GxObjectFilter* class that you can use to browse for *GxStreetMapDataset* objects using a *GxDialog* object.

The following VBA code demonstrates how to use the *GxFilterStreetMapFeatureClasses* coclass to browse for *StreetMap* feature classes. This code assumes that you have a valid *StreetMap* USA license and that the *StreetMap* USA extension is enabled.

```
Sub BrowseForStreetMapFeatureClass()
    Dim pGxDialog As IGxDialog
    Dim pGxCatalog As IGxCatalog
    Dim pGxStreetMapDataset As IGxDataset

    Set pGxDialog = New GxDialog
    With pGxDialog
        .AllowMultiSelect = False
        Set .ObjectFilter = New GxFilterStreetMapFeatureClasses
        .DoModalOpen ThisDocument.Parent.hwnd, Nothing
        Set pGxCatalog = .InternalCatalog
        Set pGxStreetMapDataset = pGxCatalog.SelectedObject
    End With
End Sub
```

Copyright © 2001 ESRI. All rights reserved. ESRI is a trademark of ESRI, registered in the United States and certain other countries; registration is pending in the European Community. *StreetMap*, *ArcCatalog*, *ArcGIS*, and *ArcMap* are trademarks of ESRI. Other companies and products mentioned herein are trademarks or registered trademarks of their respective trademark owners.